Orthogonal Polynomials and Computer Algebra

Wolfram Koepf

Abstract. Classical orthogonal polynomials of the Askey-Wilson scheme have extremely many different properties, e.g. satisfying differential equations, recurrence equations, having hypergeometric representations, Rodrigues formulas, generating functions, moment representations etc. Using computer algebra it is possible to switch between one representation and another algorithmically. Such algorithms will be discussed and implementations are presented using Maple.

Mathematics Subject Classification (2000). Primary 33C20, 33F10; Secondary 30B10, 68W30.

Keywords. Computer algebra; classical orthogonal polynomials; Askey-Wilson scheme.

1. Orthogonal Polynomials

Given: a *scalar product*

$$\langle f, g\rangle := \int_\alpha^\beta f(x)g(x)\,d\mu(x)$$

with non-negative Borel measure $\mu(x)$ supported in the interval $[\alpha, \beta]$. The following special cases are most important:

- *absolutely continuous* measure $d\mu(x) = \rho(x)\,dx$ with weight function $\rho(x)$,
- *discrete* measure $\mu(x) = \rho(x)$ supported in $\mathbb{Z}$,
- discrete measure $\mu(x) = \rho(x)$ supported in $q^{\mathbb{Z}}$.

A system of polynomials $(P_n(x))_{n\geqq 0}$

$$P_n(x) = k_n x^n + k_n' x^{n-1} + k_n'' x^{n-2} + \cdots, \quad k_n \neq 0 \tag{1.1}$$

is called *orthogonal* (OPS) w.r.t. the *positive-definite* measure $d\mu(x)$, if

$$\langle P_m, P_n \rangle = \begin{cases} 0 & \text{if } m \neq n \\ h_n > 0 & \text{if } m = n \end{cases} .$$

Using the Gram-Schmidt orthogonalization procedure one can compute the orthogonal polynomials $P_n(x)$ iteratively up to a constant standardization factor. One option is to compute the monic system with $k_n = 1$.

We define the scalar product of the Legendre polynomials $P_n(x)$:

```
> ScalarProduct:=proc(f,g,x) int(g*f,x=-1..1) end proc:
```

and declare the Gram-Schmidt procedure

```
> GramSchmidt := proc (n, x) local j, k, g, liste;
>    liste := [seq(x^j, j = 0 .. n)]; g(0) := 1;
>    for j to n do g(j):=op(j+1,liste)-
>    add(ScalarProduct(op(j+1,liste),g(k),x)*g(k)/
>    ScalarProduct(g(k), g(k), x), k = 0 .. j-1) end do;
>    [seq(g(j), j = 0 .. n)]
> end proc:
```

Now we can compute using the Gram-Schmidt procedure.

```
> SEQ1 := GramSchmidt(10, x);
```

$$\mathit{SEQ1} := [1, x, x^2 - 1/3, x^3 - 3/5\,x, x^4 + \frac{3}{35} - 6/7\,x^2, x^5 + \frac{5\,x}{21} - \frac{10\,x^3}{9},$$
$$x^6 - \frac{5}{231} + \frac{5\,x^2}{11} - \frac{15\,x^4}{11}, x^7 - \frac{35\,x}{429} + \frac{105\,x^3}{143} - \frac{21\,x^5}{13},$$
$$x^8 + \frac{7}{1287} - \frac{28\,x^2}{143} + \frac{14\,x^4}{13} - \frac{28\,x^6}{15}, x^9 + \frac{63\,x}{2431} - \frac{84\,x^3}{221}$$
$$+ \frac{126\,x^5}{85} - \frac{36\,x^7}{17}, x^{10} - \frac{63}{46189} + \frac{315\,x^2}{4199} - \frac{210\,x^4}{323} + \frac{630\,x^6}{323} - \frac{45\,x^8}{19}]$$

This computation has created the first 11 monic Legendre polynomials. Of course, *Maple* knows the Legendre polynomials internally as `LegendreP(k,x)`:

```
> SEQ2:=expand([seq(LegendreP(k,x),k=0..10)]);
```

$$\mathit{SEQ2} := [1, x, -\frac{1}{2} + \frac{3}{2}\,x^2, \frac{5}{2}\,x^3 - \frac{3}{2}\,x, \frac{3}{8} + \frac{35\,x^4}{8} - \frac{15\,x^2}{4}, \frac{63\,x^5}{8} - \frac{35\,x^3}{4} + \frac{15\,x}{8},$$
$$-\frac{5}{16} + \frac{231\,x^6}{16} - \frac{315\,x^4}{16} + \frac{105\,x^2}{16}, \frac{429\,x^7}{16} - \frac{693\,x^5}{16} + \frac{315\,x^3}{16} - \frac{35\,x}{16}, \frac{35}{128} + \frac{6435\,x^8}{128}$$
$$- \frac{3003\,x^6}{32} + \frac{3465\,x^4}{64} - \frac{315\,x^2}{32}, \frac{12155\,x^9}{128} - \frac{6435\,x^7}{32} + \frac{9009\,x^5}{64} - \frac{1155\,x^3}{32} + \frac{315\,x}{128},$$
$$-\frac{63}{256} + \frac{46189\,x^{10}}{256} - \frac{109395\,x^8}{256} + \frac{45045\,x^6}{128} - \frac{15015\,x^4}{128} + \frac{3465\,x^2}{256}]$$

Obviously the ratios of the corresponding polynomials must be constant:

```
> normal([seq(op(k,SEQ2)/op(k,SEQ1),k=1..11)]);
```

$$[1, 1, 3/2, 5/2, \frac{35}{8}, \frac{63}{8}, \frac{231}{16}, \frac{429}{16}, \frac{6435}{128}, \frac{12155}{128}, \frac{46189}{256}]$$

Every OPS has the following main properties:

- **(Three-term Recurrence)** Every OPS satisfies

$$x\,P_n(x) = a_n\,P_{n+1}(x) + b_n\,P_n(x) + c_n\,P_{n-1}(x)\ .$$

- **(Zeros)** All zeros of an OPS are simple, lie in the interior of $[\alpha, \beta]$ and have some nice interlacing properties.
- **(Hankel Matrix: Representation by Moments)**

$$P_n(x) = C_n \begin{vmatrix} \mu_0 & \mu_1 & \cdots & \mu_n \\ \mu_1 & \mu_2 & \cdots & \mu_{n+1} \\ \vdots & \vdots & \vdots & \vdots \\ \mu_{n-1} & \mu_{n+1} & \cdots & \mu_{2n-1} \\ 1 & x & \cdots & x^n \end{vmatrix},$$

where $\mu_n := \int_a^b x^n d\mu(x)$ denote the moments of $d\mu(x)$.

We can compute the moment matrix by

```
> Momentmatrix := proc (n)
> local j, k;
> convert([seq([seq(ScalarProduct(x^j, x^k, x), j = 0 .. n)],
> k = 0 .. n)], Matrix)
> end proc:
```

```
> H := Momentmatrix(5);
```

$$\begin{pmatrix} 2 & 0 & 2/3 & 0 & 2/5 & 0 \\ 0 & 2/3 & 0 & 2/5 & 0 & 2/7 \\ 2/3 & 0 & 2/5 & 0 & 2/7 & 0 \\ 0 & 2/5 & 0 & 2/7 & 0 & 2/9 \\ 2/5 & 0 & 2/7 & 0 & 2/9 & 0 \\ 0 & 2/7 & 0 & 2/9 & 0 & 2/11 \end{pmatrix}$$

and the Hankel matrix is given by

```
> Hankelmatrix := proc (n)
> local j, k, m;
> m := [seq([seq(ScalarProduct(t^j,t^k,t),j=0..n)],k=0..n-1)];
> m := [op(m), [seq(x^k, k = 0 .. n)]];
> convert(m, Matrix)
> end proc:
```

Its determinant gives a multiple of the orthogonal polynomial whose degree is the size of the square matrix minus 1. The determinant of the following matrix is therefore a multiple of $P_5(x)$:

```
> Hankelmatrix(5);
```

$$\begin{pmatrix} 2 & 0 & 2/3 & 0 & 2/5 & 0 \\ 0 & 2/3 & 0 & 2/5 & 0 & 2/7 \\ 2/3 & 0 & 2/5 & 0 & 2/7 & 0 \\ 0 & 2/5 & 0 & 2/7 & 0 & 2/9 \\ 2/5 & 0 & 2/7 & 0 & 2/9 & 0 \\ 1 & x & x^2 & x^3 & x^4 & x^5 \end{pmatrix}$$

and the first 6 multiples of the Legendre polynomials are given by

```
> SEQ3:=[seq(LinearAlgebra[Determinant](Hankelmatrix(n)),n=0..5)];
```

$$\mathit{SEQ3} := \Big[1, 2\,x, 4/3\,x^2 - 4/9, \frac{32\,x^3}{135} - \frac{32\,x}{225},$$
$$\frac{256\,x^4}{23625} - \frac{512\,x^2}{55125} + \frac{256}{275625}, \frac{32768\,x^5}{260465625} - \frac{65536\,x^3}{468838125} + \frac{32768\,x}{1093955625}\Big]$$

Again, the ratios of the corresponding polynomials must be constant:

```
> normal([seq(op(k,SEQ3)/op(k,SEQ1),k=1..6)]);
```

$$[1, 2, 4/3, \frac{32}{135}, \frac{256}{23625}, \frac{32768}{260465625}]$$

2. Classical Orthogonal Polynomials

The *classical* OPS $(P_n(x))_{n\geqq 0}$ can be defined as the polynomial solutions of the *differential equation*:

$$\sigma(x)P_n''(x) + \tau(x)P_n'(x) - \lambda_n P_n(x) = 0\ . \tag{2.1}$$

Substituting (1.1) into (2.1), we conclude:

- $n = 1$ yields $\tau(x) = dx + e, d \neq 0$,
- $n = 2$ yields $\sigma(x) = ax^2 + bx + c$,
- The coefficient of x^n yields $\lambda_n = n(a(n-1) + d)$.

These classical families can be classified (modulo linear transformations) according to the following scheme (Bochner (1929), [2])

- $\sigma(x) = 0$ powers x^n,
- $\sigma(x) = 1$ *Hermite* polynomials,
- $\sigma(x) = x$ *Laguerre* polynomials,
- $\sigma(x) = 1 - x^2$ *Jacobi* polynomials,
- $\sigma(x) = x^2$ *Bessel* polynomials.

For the theory one needs

- a representing basis $f_n(x)$, here the powers $f_n(x) = x^n$;
- an operator, here the derivative operator D, with $D\, f_n(x) = n\, f_{n-1}(x)$.

The corresponding weight function $\rho(x)$ satisfies the *Pearson Differential Equation*

$$\frac{d}{dx}\Big(\sigma(x)\rho(x)\Big) = \tau(x)\rho(x)\ . \tag{2.2}$$

Hence the weight function is given by

$$\rho(x) = \frac{C}{\sigma(x)} e^{\int \frac{\tau(x)}{\sigma(x)} dx} .$$

The following properties are equivalent, each defining the classical continuous families:

- Differential equation (2.1) for $(P_n(x))_{n \geqq 0}$.
- Pearson differential equation (2.2) $(\sigma\,\rho)' = \tau\,\rho$ for the weight $\rho(x)$.
- With $(P_n(x))_{n \geqq 0}$ also $(P'_{n+1}(x))_{n \geqq 0}$ is an OPS.
- **Derivative Rule**:

$$\sigma(x)\,P'_n(x) = \alpha_n\,P_{n+1}(x) + \beta_n\,P_n(x) + \gamma_n\,P_{n-1}(x) .$$

- **Structure Relation**: $P_n(x)$ satisfies

$$P_n(x) = \widehat{a}_n\,P'_{n+1}(x) + \widehat{b}_n\,P'_n(x) + \widehat{c}_n\,P'_{n-1}(x) .$$

- **Rodrigues Formula**: $P_n(x)$ is given as

$$P_n(x) = \frac{E_n}{\rho(x)} \frac{d^n}{dx^n} \Big(\rho(x)\,\sigma(x)^n\Big) .$$

3. Classical Discrete Orthogonal Polynomials

The *classical discrete* OPS can be analogously defined as the solutions of the *difference equation* (Lesky (1962), [12]):

$$\sigma(x)\Delta\nabla P_n(x) + \tau(x)\Delta P_n(x) - \lambda_n P_n(x) = 0 \qquad (3.1)$$

where $\Delta f(x) = f(x+1) - f(x)$ and $\nabla f(x) = f(x) - f(x-1)$ denote the forward and backward difference operators. As in the continuous case, we get

- $n = 1$ yields $\tau(x) = dx + e, d \neq 0$,
- $n = 2$ yields $\sigma(x) = ax^2 + bx + c$,
- The coefficient of x^n yields $\lambda_n = n(a(n-1) + d)$.

The classical discrete families can be classified (modulo linear transformations) according to the following scheme (Lesky (1962), [12], see also, [14]):

- $\sigma(x) = 0$ falling factorials $x^{\underline{n}} = x(x-1)\cdots(x-n+1)$,
- $\sigma(x) = 1$ shifted **Charlier** polynomials,
- $\sigma(x) = x$ **Charlier**, **Meixner**, **Krawtchouk** polynomials,
- $\deg(\sigma(x), x) = 2$ **Hahn** polynomials.

For the theory one needs

- a representing basis $f_n(x)$, here the falling factorial $f_n(x) = x^{\underline{n}}$;
- an operator, here the operator Δ, with $\Delta\,f_n(x) = n\,f_{n-1}(x)$.

The corresponding discrete weight function $\rho(x)$ satisfies the Pearson difference equation

$$\Delta\Big(\sigma(x)\rho(x)\Big) = \tau(x)\rho(x) .$$

Hence it is given by the term ratio

$$\frac{\rho(x+1)}{\rho(x)} = \frac{\sigma(x)+\tau(x)}{\sigma(x+1)} \ . \tag{3.2}$$

We would like to put our results into the general framework of hypergeometric functions.

The power series

$${}_pF_q\left(\begin{array}{c} a_1,\ldots,a_p \\ b_1,\ldots,b_q \end{array}\middle|\, z\right) = \sum_{k=0}^{\infty} A_k\, z^k \ ,$$

whose summands $\alpha_k = A_k z^k$ have a rational term ratio

$$\frac{\alpha_{k+1}}{\alpha_k} = \frac{A_{k+1}\, z^{k+1}}{A_k\, z^k} = \frac{(k+a_1)\cdots(k+a_p)}{(k+b_1)\cdots(k+b_q)}\,\frac{z}{(k+1)} \ ,$$

is called the *generalized hypergeometric series*. The summand $\alpha_k = A_k z^k$ of a hypergeometric series is called a *hypergeometric term*.

The relation (3.2) therefore tells that the weight function $\rho(x)$ of the classical discrete orthogonal polynomials is a hypergeometric term w.r.t. the variable x.

For the coefficients of the generalized hypergeometric series one gets the following formula

$${}_pF_q\left(\begin{array}{c} a_1,\ldots,a_p \\ b_1,\ldots,b_q \end{array}\middle|\, z\right) = \sum_{k=0}^{\infty} \frac{(a_1)_k \cdots (a_p)_k}{(b_1)_k \cdots (b_q)_k}\,\frac{z^k}{k!}$$

using the *Pochhammer symbol* $(a)_k = a(a+1)\cdots(a+k-1) = \frac{\Gamma(a+k)}{\Gamma(a)}$.

From the differential equation (2.1) one can compute a recurrence equation for the corresponding power series coefficients [16]. Using *Maple*, we get

```
> sigma := a*x^2+b*x+c; tau := d*x+e;
```

$$\sigma := ax^2 + xb + c$$
$$\tau := dx + e$$

```
> DE := sigma*(diff(F(x), x$2))+tau*(diff(F(x), x))-n*(a*n-a+d)*F(x);
```

$$DE := \left(ax^2+xb+c\right)\frac{\mathrm{d}^2}{\mathrm{d}x^2}F(x) + (dx+e)\frac{\mathrm{d}}{\mathrm{d}x}F(x) - n\,(an-a+d)\,F(x)$$

This differential equation is converted towards the recurrence equation

```
> RE := gfun[diffeqtorec](DE, F(x), A(k));
```

$$\begin{aligned} RE := {} & \left(ak^2 + (-a+d)\ k - an^2 + an - dn\right) A(k) \\ & + \left(bk^2 + (b+e)\ k + e\right) A(k+1) + \left(ck^2 + 3\,c\,k + 2\,c\right) A(k+2) \end{aligned}$$

The Laguerre polynomials have the data

```
> laguerre := {a = 0, b = 1, c = 0, d = -1, e = alpha+1};
```

$$\mathit{laguerre} := \{a=0, b=1, c=0, d=-1, e=\alpha+1\}$$

so that we get for their power series coefficients A_k

```
> laguerreRE := subs(laguerre, RE);
```

$$laguerreRE := (-k+n)\,A(k) + \left(k^2\,(2+\alpha)\;k+\alpha+1\right) A(k+1)$$

Therefore their quotient A_{k+1}/A_k is given in factored form by

```
> quotient := factor(solve(laguerreRE, A(k+1))/A(k));
```

$$quotient := \frac{k-n}{(k+1)\,(k+\alpha+1)}$$

from which one can read off directly the hypergeometric representation

```
> lag := hypergeom([-n], [alpha+1], x);
```

$$lag := {}_1\mathrm{F}_1(-n;\,\alpha+1;\,x)$$

By an internal command, *Maple* can convert this towards

```
> convert(lag, StandardFunctions);
```

$$\frac{\Gamma(n+1)\,\Gamma(\alpha+1)\,LaguerreL(n,\alpha,x)}{\Gamma(n+\alpha+1)}$$

back, again. We have therefore seen that using this approach one gets for the Laguerre polynomials

$$L_n^{\alpha}(x) = \binom{n+\alpha}{n}\,{}_1F_1\left(\left.\begin{array}{c}-n\\ \alpha+1\end{array}\right|\,x\right) = \sum_{k=0}^{n}\frac{(-1)^k}{k!}\binom{n+\alpha}{n-k}x^k\,.$$

Another example are the Hahn polynomials which are given by

$$Q_n^{(\alpha,\beta)}(x,N) = {}_3F_2\left(\left.\begin{array}{c}-n,-x,n+1+\alpha+\beta\\ \alpha+1,-N\end{array}\right|\,1\right)\,.$$

Similarly, all the other classical systems have a hypergeometric representation. These can be found in every book about OPS, e.g. in [7], and on the CAOP web page [8].

4. Classical q-Orthogonal Polynomials and the Askey-Wilson Scheme

The *classical* q-OPS can be analogously defined as the polynomial solutions of the *q-difference equation* (Hahn (1949), [6]):

$$\sigma(x)D_qD_{1/q}P_n(x)+\tau(x)D_qP_n(x)-\lambda_{n,q}P_n(x)=0 \tag{4.1}$$

where $D_qf(x)=\frac{f(x)-f(qx)}{(1-q)x}$ is the Hahn operator. As before, we can conclude

- $n=1$ yields $\tau(x)=dx+e, d\neq 0$,
- $n=2$ yields $\sigma(x)=ax^2+bx+c$,
- The coefficient of x^n yields $\lambda_{n,q}=[n]_q(a[n-1]_q+d[n]_q)$ where $[n]_q=\frac{1-q^n}{1-q}$ is the q-bracket.

The classical q-discrete families of the Hahn class considered can be classified (modulo linear transformations) according to the following list: *Big q-Jacobi polynomials, q-Hahn polynomials, Big q-Laguerre polynomials, Al-Salam-Carlitz I polynomials, discrete q-Hermite I polynomials, Little q-Jacobi polynomials, alternative q-Charlier polynomials, Little q-Laguerre polynomials, q-Meixner polynomials, Stieltjes-Wigert polynomials, q-Laguerre polynomials, q-Charlier polynomials, Al-Salam-Carlitz II polynomials, discrete q-Hermite II polynomials*, see e.g. [7] and [8].

For the theory one needs:

- two representing bases $f_n(x)$, here $f_n(x) = x^n$ and $g_n(x) = (x;q)_n$ where $(x;q)_n = (1-x)(1-xq)\cdots(1-xq^{n-1})$ is the *q-Pochhammer symbol*;
- an operator, here the operator D_q, with $D_q\, f_n(x) = [n]_q\, f_{n-1}(x)$ and a similar relation for $g_n(x)$.

The corresponding q-discrete weight function $\rho(x)$ satisfies the Pearson q-difference equation

$$D_q\Big(\sigma(x)\rho(x)\Big) = \tau(x)\rho(x)\ .$$

Hence it is given by the term ratio:

$$\frac{\rho(qx)}{\rho(x)} = \frac{\sigma(x) + (q-1)x\tau(x)}{\sigma(qx)}\ . \tag{4.2}$$

The power series

$${}_r\phi_s\left(\begin{array}{c} a_1, \ldots, a_r \\ b_1, \ldots, b_s \end{array} \middle|\, q\, ;\, z\right) = \sum_{k=0}^{\infty} A_k\, z^k\ ,$$

whose summands $\alpha_k = A_k z^k$ are given by

$$A_k\, z^k = \frac{(a_1;q)_k \cdots (a_r;q)_k}{(b_1;q)_k \cdots (b_s;q)_k} \frac{z^k}{(q;q)_k} \Big((-1)^k q^{\binom{k}{2}}\Big)^{1+s-r}$$

is called the *basic hypergeometric series*. The summand $\alpha_k = A_k z^k$ of a basic hypergeometric series is called a *q-hypergeometric term*.

The relation (4.2) therefore tells that the weight function $\rho(x)$ of the classical q-orthogonal polynomials is a q-hypergeometric term w.r.t. the variable x.

In CAOP [8] you saw all the families of the *Askey-Wilson Scheme*. This scheme contains

- continuous measures supported in an interval (classical continuous OPS);
- discrete measures supported in $\mathbb{Z}$ (classical discrete OPS);
- discrete measures supported in $q^{\mathbb{Z}}$ (**Hahn tableau**);
- discrete measures supported on a quadratic lattice (**Wilson tableau**);
- discrete measures supported on a q-quadratic lattice (**Askey-Wilson tableau**).

It turns out that the last two classes can be treated in a similar way as the continuous and the discrete cases resulting in a similar theory [3].

5. Computer Algebra Applied to Classical Orthogonal Polynomials

Using linear algebra one can compute the coefficients of the following identities—expressed through the parameters a, b, c, d and e from the defining equations (2.1), (3.1) or (4.1)—(Lesky (1985), [13]):

(RE) $$x\,P_n(x) = a_n\,P_{n+1}(x) + b_n\,P_n(x) + c_n\,P_{n-1}(x)$$

(DR) $$\sigma(x)\,P_n'(x) = \alpha_n\,P_{n+1}(x) + \beta_n\,P_n(x) + \gamma_n\,P_{n-1}(x)$$

(SR) $$P_n(x) = \widehat{a}_n\,P_{n+1}'(x) + \widehat{b}_n\,P_n'(x) + \widehat{c}_n\,P_{n-1}'(x)$$

We define $P_n(x)$, given by (1.1) in *Maple*, and substitute the three highest coefficients—which will be sufficient for our purposes—into the differential equation:

```
> p := k[n]*x^n+kprime[n]*x^(n-1)+kprimeprime[n]*x^(n-2);
```

$$p := k_n x^n + kprime_n x^{n-1} + kprimeprime_n x^{n-2}$$

```
> DE := sigma*(diff(p, x$2))+tau*(diff(p, x))-lambda[n]*p:
```

We divide by x^{n-4} so that the result is a polynomial of degree 4 (whose three highest coefficients are those of x^4, x^3 and x^2)

```
> de := collect(simplify(DE/x^(n-4)), x):
```

Equating the highest coefficients yields the relation for λ_n that we already met:

```
> rule1 := lambda[n] = solve(coeff(de, x, 4), lambda[n]);
```

$$rule1 := \lambda_n = n\,(an - a + d)$$

Next, we substitute λ_n into the differential equation and equate the next highest coefficient. This shows that the second highest coefficient k_n' of $P_n(x)$ is a rational multiple of the leading coefficient k_n:

```
> de := expand(subs(rule1, de)):
> rule2 := kprime[n] = solve(coeff(de, x, 3), kprime[n]);
```

$$rule2 := kprime_n = \frac{n k_n\,(bn - b + e)}{2\,an - 2\,a + d}$$

In the last step we deduce that generically k_n'' is also a rational multiple of k_n:

```
> rule3 := kprimeprime[n] =
> solve(coeff(subs(rule2, de), x, 2), kprimeprime[n]);
```

$$rule3 := kprimeprime_n = k_n\,n \cdot \frac{\left(b^2n^3+2\,acn^2-4b^2n^2+2ben^2-4acn+5b^2n-5ben+cdn+e^2n+2ac-2b^2+3be-cd-e^2\right)}{2\,(2\,an-2\,a+d)\,(2\,an-3\,a+d)}$$

In the sequel, we consider without loss of generality the monic case.

```
> k[n] := 1;
```

$$k_n := 1$$

To get information about the coefficients of the recurrence equation, we put it in the following form to be zero.

```
> RE := x*P(n)-a[n]*P(n+1)-b[n]*P(n)-c[n]*P(n-1);
```

$$RE := xP(n) - a_n P(n+1) - b_n P(n) - c_n P(n-1)$$

After substituting $P_n(x)$, given by (1.1), and the previous results about k_n' (`rule2`) and k_n'' (`rule3`), and by equating again the three highest coefficients, we get:

```
> RE:=subs({P(n)=p,P(n-1)=subs(n=n-1,p),P(n+1)=subs(n=n+1,p)}, RE):
> RE:=subs({rule2,rule3,subs(n=n-1,rule2),subs(n=n-1,rule3),
> subs(n = n+1, rule2), subs(n = n+1, rule3)}, RE):
> re := simplify(numer(normal(RE))/x^(n-3)):
> rule4 := a[n] = solve(coeff(re, x, 4), a[n]);
```

$$rule4 := a_n = 1$$

```
> rule5 := b[n] = factor(solve(subs(rule4, coeff(re, x, 3)), b[n]));
```

$$rule5 := b_n = -\frac{2\,abn^2 - 2\,abn + 2\,bdn - 2\,ae + de}{(2\,an - 2\,a + d)\,(2\,an + d)}$$

```
> rule6 := c[n] =
> factor(solve(subs(rule5, subs(rule4, coeff(re, x, 2))), c[n]));
```

$$rule6 := c_n = -\frac{n\,(an - 2\,a + d)}{(2\,an - 2\,a + d)^2\,(2\,an - 3\,a + d)\,(2\,an - a + d)} \cdot \left(4a^2cn^2 - ab^2n^2 - 8a^2cn + 2ab^2n + 4acdn - b^2dn + 4a^2c - ab^2 - 4acd + ae^2 + b^2d - bde + cd^2\right)$$

Such relations were given generically in the paper [9] for the continuous and the discrete cases, and in later papers extended to the q-case ([10], [4]) and to the quadratic case ([5], [15], [17]).

They can be used to compute power series coefficients, inversion coefficients, connection coefficients and parameter derivatives, as e.g. [9]

$$\frac{\partial}{\partial\alpha} L_n^{(\alpha)}(x) = \sum_{m=0}^{n-1} \frac{1}{n-m} L_m^{(\alpha)}(x)\ .$$

We have shown that the coefficients of the recurrence equation of the classical systems can be written in terms of the coefficients a, b, c, d, and e of the differential / difference equation.

If one uses these formulas in the backward direction, then one can determine the possible differential / difference equations from a given recurrence. For this purpose one must solve a *non-linear system*.

Assume the following recurrence equation is given:

$$P_{n+2}(x) - (x - n - 1)\,P_{n+1}(x) + \alpha(n+1)^2 P_n(x) = 0\ .$$

Does this equation have classical OPS solutions?

We find out [10] that the solutions of this equation are shifted Laguerre polynomials for $\alpha = 1/4$. For $\alpha < 1/4$ the recurrence has Meixner and Krawtchouk polynomial solutions.

```
> read "hsum17.mpl";
   ‘Package "Hypergeometric Summation", Maple V - Maple 17‘
  ‘Copyright 1998-2013, Wolfram Koepf, University of Kassel‘
> read "retode.mpl";
            ‘Package "REtoDE", Maple V - Maple 8‘
  ‘Copyright 2000-2002, Wolfram Koepf, University of Kassel‘
> RE := P(n+2)-(x-n-1)*P(n+1)+alpha*(n+1)^2*P(n) = 0;
```

$$RE := P(n+2) - (x-n-1)\,P(n+1) + \alpha\,(n+1)^2\,P(n) = 0$$

```
> REtoDE(RE, P(n), x);
  ‘Warning: parameters have the values‘
```

$\{a = 0, \alpha = 1/4, b = -d/2, c = -d/4, d = d, e = 0\}$

$$\left[1/2\,(2\,x+1)\,\frac{\partial^2}{\partial x^2}P(n,x) - 2\,x\frac{\partial}{\partial x}P(n,x) + 2\,nP(n,x) = 0,\right.$$
$$\left.\left[I = [-1/2,\infty], \rho(x) = 2\,\mathrm{e}^{-2\,x}\right], \frac{k_{n+1}}{k_n} = 1\right]$$

```
> REtodiscreteDE(RE, P(n), x);
```

For the last computation we omit the lengthy output and just state that the difference equation and weight of the Meixner and Krawtchouk polynomials is discovered.

Recently Walter Van Assche asked me the question to find all OPS of the Askey-Wilson scheme that satisfy a certain recurrence equation, see [19]? Dr. Daniel Tcheutia solved this question completely [18] by extending the shown algorithm to the quadratic lattice. The answer is: The adapted algorithm finds the solutions to the first question (that were already know to Walter van Assche). This algorithm also *proves that the second recurrence equation does not have such solutions.*

The *Legendre Polynomials* $P_n(x)$ of the Jacobi class have several representations as series:

$$\begin{aligned}
P_n(x) &= \sum_{k=0}^{n} \binom{n}{k}\binom{-n-1}{k}\left(\frac{1-x}{2}\right)^k \\
&= \frac{1}{2^n}\sum_{k=0}^{n}\binom{n}{k}^2 (x-1)^{n-k}(x+1)^k \\
&= \frac{1}{2^n}\sum_{k=0}^{\lfloor n/2\rfloor}(-1)^k\binom{n}{k}\binom{2n-2k}{n}x^{n-2k}\,.
\end{aligned}$$

It is already non-trivial to identify that these three series represent the same functions, but Zeilberger's algorithm [11] computes the desired normal forms, namely the corresponding (and identical) recurrence equations jointly with enough initial values.

```
> legendreterm1:=binomial(n,k)*binomial(-n-1,k)*((1-x)*(1/2))^k;
```

$$\mathit{legendreterm1} := \binom{n}{k}\binom{-n-1}{k}(1/2-x/2)^k$$

```
> legendreterm2 := binomial(n, k)^2*(x-1)^(n-k)*(x+1)^k/2^n;
```

$$\mathit{legendreterm2} := \frac{\left(\binom{n}{k}\right)^2 (x-1)^{-k+n}(x+1)^k}{2^n}$$

```
> legendreterm3 :=
> (-1)^k*binomial(n, k)*binomial(2*n-2*k, n)*x^(n-2*k)/2^n;
```

$$\mathit{legendreterm3} := \frac{(-1)^k \binom{n}{k}\binom{2n-2k}{n} x^{n-2k}}{2^n}$$

```
> RE1 := {sumrecursion(legendreterm1, k, P(n)),
> P(0) = add(subs(n = 0, legendreterm1), k = 0 .. 0),
> P(1) = add(subs(n = 1, legendreterm1), k = 0 .. 1)};
```

$\mathit{RE1} :=$

$$\{(n+2)\,P(n+2) - x\,(2\,n+3)\,P(n+1) + (n+1)\,P(n) = 0, P(0) = 1, P(1) = x\}$$

```
> RE2 := {sumrecursion(legendreterm2, k, P(n)),
> P(0) = add(subs(n = 0, legendreterm2), k = 0 .. 0),
> P(1) = add(subs(n = 1, legendreterm2), k = 0 .. 1)};
```

$\mathit{RE2} :=$

$$\{(n+2)\,P(n+2) - x\,(2\,n+3)\,P(n+1) + (n+1)\,P(n) = 0, P(0) = 1, P(1) = x\}$$

```
> RE3 := {sumrecursion(legendreterm3, k, P(n)),
> P(0) = expand(add(subs(n = 0, legendreterm3), k = 0 .. 0)),
> P(1) = expand(add(subs(n = 1, legendreterm3), k = 0 .. 1))};
```

$\mathit{RE3} :=$

$$\{(n+2)\,P(n+2) - x\,(2\,n+3)\,P(n+1) + (n+1)\,P(n) = 0, P(0) = 1, P(1) = x\}$$

The above computations have computed the normal forms of each of the three different series representations. Since they agree, we have proved that the series represent the same family of functions.

Next, we compute their hypergeometric representations.

```
> Sumtohyper(legendreterm1, k);
```

$$\mathit{Hypergeom}\left([n+1,-n],[1],1/2-x/2\right)$$

```
> Sumtohyper(legendreterm2, k);
```

$$\frac{(x-1)^n}{2^n}\mathit{Hypergeom}\left([-n,-n],[1],\frac{x+1}{x-1}\right)$$

```
> convert(Sumtohyper(legendreterm3, k), binomial);
```

$$\frac{x^n\, \mathit{Hypergeom}\left([-n/2, 1/2-n/2], [-n+1/2], x^{-2}\right) \binom{2\,n}{n}}{2^n}$$

It can also be easily shown that they satisfy the same differential equation.

```
> DE1 := sumdiffeq(legendreterm1, k, P(x));
```

$$\mathit{DE1} := \left(\frac{d^2}{dx^2} P(x)\right)(x-1)(x+1) + 2\,x \frac{d}{dx} P(x) - n\,(n+1)\,P(x) = 0$$

```
> DE2 := sumdiffeq(legendreterm2, k, P(x));
```

$$\mathit{DE2} := \left(\frac{d^2}{dx^2} P(x)\right)(x-1)(x+1) + 2\,x \frac{d}{dx} P(x) - n\,(n+1)\,P(x) = 0$$

```
> DE3 := sumdiffeq(legendreterm3, k, P(x));
```

$$\mathit{DE3} := \left(\frac{d^2}{dx^2} P(x)\right)(x-1)(x+1) + 2\,x \frac{d}{dx} P(x) - n\,(n+1)\,P(x) = 0$$

In the talk given by Naoures Ayadi [1], she introduced the Meixner type polynomials

$$\widehat{M}_n^{\beta_1,\beta_2}(x,c) = (\beta_1)_k\,(\beta_2)_k\,{}_2F_2\left(\begin{array}{c} -n, -x \\ \beta_1, \beta_2 \end{array} \middle| \begin{array}{c} \frac{1}{c} \end{array}\right).$$

Using Zeilberger's algorithm, it is easy to get a recurrence equation for $M_n^{\beta_1,\beta_2}(x,c)$.

```
> meixnersummand := pochhammer(beta[1], n)*pochhammer(beta[2], n)*
> hyperterm([-n, -x], [beta[1], beta[2]], 1/c, k);
```

$$\mathit{meixnersummand} := \frac{\mathit{pochhammer}(\beta_1, n)\,\mathit{pochhammer}(\beta_2, n)\,\mathit{pochhammer}(-n, k)\,\mathit{pochhammer}(-x, k)\left(c^{-1}\right)^k}{\mathit{pochhammer}(\beta_1, k)\,\mathit{pochhammer}(\beta_2, k)\,k!}$$

```
> MeixnerRE := sumrecursion(meixnersummand, k, M(n));
```

$$\begin{aligned} \mathit{MeixnerRE} &:= -cM(n+3) \\ &+ \left(3\,cn^2 + 2\,cn\beta_1 + 2\,cn\beta_2 + c\beta_1\beta_2 + 11\,cn + 4\,c\beta_1 + 4\,c\beta_2 + 10\,c - n + x - 2\right) M(n+2) \\ &- (n+2)(1+\beta_2+n)(1+\beta_1+n)(3\,cn + c\beta_1 + c\beta_2 + 4\,c - 1)\,M(n+1) \\ &+ c\,(n+2)(n+1)(1+\beta_2+n)(\beta_2+n)(1+\beta_1+n)(\beta_1+n)\,M(n) = 0 \end{aligned}$$

The more complicated family

$$M_n^{\beta_1,\beta_2,\beta_3}(x,c) = (\beta_1)_k\,(\beta_2)_k\,(\beta_3)_k\,{}_2F_3\left(\begin{array}{c} -n, -x \\ \beta_1, \beta_2, \beta_3 \end{array} \middle| \begin{array}{c} \frac{1}{c} \end{array}\right)$$

is similarly feasible.

6. Epilogue

Software developers love when their software is used. But they need your support. Hence my suggestion: If you use one of the packages mentioned for your scientific work, please cite its use!

References

[1] Ayadi, N.: Classical discrete d-orthogonal polynomials. Contributed Talk 27 at the AIMS-Volkswagen Stiftung Workshop on Introduction to Orthogonal Polynomials and Applications, Douala, Cameroon, October 5–12, 2018. `http://www.aims-volkswagen-workshops.org/workshop-information.html`

[2] Bochner, S.: Über Sturm-Liouvillesche Polynomsysteme. Math. Z. **29**, 1929, 730–736.

[3] Foupouagnigni, M.: On difference equations for orthogonal polynomials on nonuniform lattices. J. Difference Equ. Appl. **14**, 2008, 127–174.

[4] Foupouagnigni, M., Koepf, W., Tcheutia, D. D. and Njionou Sadjang, P.: Representations of q-orthogonal polynomials. J. Symbolic Comput. **47**, 2012, 1347–1371.

[5] Foupouagnigni, M., Koepf, W. and Tcheutia, D. D.: Connection and linearization coefficients of the Askey-Wilson polynomials. J. Symbolic Computation **53**, 2013, 96–118.

[6] Hahn, W.: Über Orthogonalsysteme, die q-Differenzengleichungen genügen. Math. Nachr. **2**, 1949, 4–34.

[7] Koekoek, R., Lesky, P. and Swarttouw, R. F.: *Hypergeometric Orthogonal Polynomials and their q-analogues*. Springer Monographs in Mathematics, Springer, Berlin, 2010.

[8] Koepf, W.: *Computer Algebra and Orthogonal Polynomials*: `www.caop.org`.

[9] Koepf, W. and Schmersau, D.: Representations of orthogonal polynomials. J. Comput. Appl. Math. **90**, 1998, 57–94.

[10] Koepf, W. and Schmersau, D.: Recurrence equations and their classical orthogonal polynomial solutions. Appl. Math. Comput. **128**, 2002, 303–327.

[11] Koepf, W.: *Hypergeometric Summation. An Algorithmic Approach to Summation and Special Function Identities*. First Edition, Vieweg, Braunschweig–Wiesbaden, 1998; Second Edition, Springer Universitext. Springer, London, 2014.

[12] Lesky, P.: Über Polynomsysteme, die Sturm-Liouvilleschen Differenzengleichungen genügen. Math. Z. **78**, 1962, 439–445.

[13] Lesky, P.: Über Polynomlösungen von Differewntialgleichungen und Differenzengleichungen zweiter Ordnung. Anzeiger der Österreichischen Akademie der Wissenschaften, math.-naturwiss. Klasse **121**, 1985, 29–33.

[14] Nikiforov, A. F., Suslov, S. K. and Uvarov, V. B.: *Classical Orthogonal Polynomials of a Discrete Variable*. Springer, Berlin–Heidelberg–New York, 1991.

[15] Njionou Sadjang, P., Koepf, W. and Foupouagnigni, M.: On structure formulas for Wilson polynomials. Integral Transforms and Special Functions **26**, 2015, 1000–1014.

[16] Salvy, B. and Zimmermann, P.: GFUN: A Maple package for the manipulation of generating and holonomic functions in one variable. ACM Transactions on Mathematical Software **20**, 1994, 163–177.

[17] Tcheutia, D. D., Njionou Sadjang, P., Koepf, W. and Foupouagnigni, M.: Divided-difference equation, inversion, connection, multiplication and linearization formulae of the continuous Hahn and the Meixner-Pollaczek Polynomials. Ramanujan Journal **45**, 2018, 33–56.

[18] Tcheutia, D.: Recurrence equations and their classical orthogonal polynomial solutions on a quadratic or q-quadratic lattice. J. Difference Equ. Appl. DOI: 10.1080/10236198.2019.1627346, 2019.

[19] Van Assche, Walter: Solution of an open problem about two families of orthogonal polynomials. SIGMA **15**, 2019, 005, 6 pages.

Wolfram Koepf
Institute of Mathematics
University of Kassel
Heinrich-Plett-Str. 40
D–34132 Kassel, Germany
e-mail: `koepf@mathematik.uni-kassel.de`